Hackers Guide to Visual FoxPro 6.0

S4G352. Readings on DELETE-SQL.

Delete-SQL

This is the "other DELETE." Like the Xbase DELETE, it lets you mark one or more records in a table for deletion. The records aren't physically deleted until you PACK the table.

	Usage
	DELETE FROM [ Database! ] Table

     [ WHERE lCondition ]


	Parameter
	Value
	Meaning

	Database
	Name
	Name of database containing the table from which records are to be deleted.

	
	Omitted 
	Delete records from a free table or the named table in the current database.

	Table
	Name
	Table from which to delete records.

	lCondition
	Logical
	Determines which records are deleted.

	
	Omitted
	All records in cTable are deleted.


Be careful with this command. Unlike the Xbase DELETE, the default for DELETE-SQL is to delete all records in the table.

	Even when there's a tag on DELETED(), DELETE-SQL doesn't check the tag unless you include the clause NOT DELETED() in the WHERE clause. Amazing. If the clause is included, you get a Rushmore-ized delete.


Because the two-step delete is an Xbase concept, there's no SQL equivalent to RECALL. You have to use the Xbase RECALL command.

Despite its SQL antecedents, DELETE-SQL only lets you delete records in a single table. Because you can list only one table in the FROM clause, there's no way to create a join condition with another table. However, you can use a sub-query in the WHERE clause to base the deletion decision on information from other tables. Functionally, DELETE-SQL doesn't provide anything that Xbase DELETE doesn't already have. The only slight advantage it offers is that the table doesn't need to be open initially, but it leaves it open afterward anyway.

You might think it would be better to use DELETE-SQL when dealing with remote data. But the truth is that you'll always be working on a view of that data, and Xbase DELETE works just as well. We're willing to have our opinion changed, but right now we can't see a whole lot of reasons to use DELETE-SQL.

	Example
	* delete all orders for a specified customer

DELETE FROM TasTrade!Orders WHERE customer_id="WOLZA"

* delete all orders for the current customer record

* assumes Customer is open 

DELETE FROM TasTrade!Orders ;

   WHERE Customer_id=Customer.Customer_id

* Get rid of customers who've never bothered to place an order

DELETE FROM TasTrade!Customer ;

   WHERE Customer_id NOT IN ;

         (SELECT Customer_ID FROM Orders)


	See Also
	Delete, Deleted(), Recall, Sys(3054)


S4G490. Readings on Container, Control.

Container, Control

These two base classes with the incredibly confusing names are really just empty shells. The difference between them is that Container's shell is transparent, while Control's is opaque. (We'll handle the name conflict by referring to these classes with capital "C" and to the similarly named concepts with small "c".)

Both of these classes are containers (note the small "c" here) that let you combine other controls (small "c" here, too) into a unified whole. Container gives you access to the individual objects inside, while Control hides those objects from others. Essentially, Control protects the objects inside from meddlers. In either case, though, you can access the PEMs of the resulting object.

You'll use these classes as the foundation for building things you wish were in the product in the first place. For example, the date spinner class DateSpin in the CoolStuf library on the CD is really a sub-class of Container. Now that we've done the work for you, though, you can just drop a date spinner onto any form and use it as if it were a native control.

There's a third similar base class, Custom, which lets you construct non-visual objects. Like Container, it gives you access to the things inside.

Container and Control have few PEMs of their own because there's not much there until you put it there. With these base classes, you're definitely likely to add custom properties and methods when you sub-class.

	Property
	Value
	Purpose

	ControlCount
	Numeric
	The number of controls inside.

	Controls
	Collection
	References to the controls inside.

	Picture
	Character
	Name of a bitmap file that is tiled to produce "wallpaper" for the control.


	Method
	Purpose

	AddObject, RemoveObject
	Container only. These methods let you add and remove objects from the container at runtime. Control doesn't support them because you can't do this, since the controls are not exposed.


The different behavior of Container and Control carries through to subclassing as well. When you subclass something based on Container, you can change the things inside the original Container. When you subclass an object based on Control, the objects inside don't even show up in the property sheet. The class is truly a black box.

	Example
	* See the DateSpin example on the CD.

* Here's the beginning of the definition 

* (as output by the Class Browser 

*  and prettied up just a little).

DEFINE CLASS datespin AS container

   Width = 176

   Height = 31

   BackStyle = 0

   BorderWidth = 1

   *-- Currently chosen month

   nMonth = 1

   *-- Currently Chosen Day

   nDay = 1

   *-- Currently Chosen Year

   nYear = 1900

   Name = "datespin"

   *-- Contain the date currently displayed

   Value = .F.

ADD OBJECT label1 AS label WITH ;

      Comment = "/", ;

      FontBold = .F., ;

      FontSize = 24, ;

      BackStyle = 0, ;

      Caption = "/", ;

      Height = 18, ;

      Left = 48, ;

      Top = 5, ;

      Width = 13, ;

      Name = "Label1"


	See Also
	AddObject, ControlCount, Controls, Custom, Picture, RemoveObject


S4G738. Readings on SCCProvider.

SCCProvider

This property of the Project object tells you whether the project is under source code control, and if so, who the provider is.

	Usage
	cProvider = prjProject.SCCProvider


The SCCProvider property returns, as a string, the Registry key under HKEY_LOCAL_MACHINE that contains information about the source code control provider for this project. You can use this information to plunge into the Registry and get the settings for this particular provider. As we haven't played with anything other than Visual SourceSafe, we can't tell you much about what you can find there, but you'll probably want to fire up a good Registry editor and see what you can find. 

	Example
	#DEFINE HKEY_LOCAL_MACHINE   0x80000002

lcProvider = oProject.SCCProvider

IF NOT EMPTY(lcProvider)

   oRegistry = NewObject("Registry", ;

                          HOME() + "FFC\Registry.vcx")

   cValue = space(255)

   oRegistry.GetRegKey("SCCServerName ", @cValue, ;

                    lcProvider, HKEY_LOCAL_MACHINE)

   ? "The project's database is controlled by " + cValue

ENDIF


	See Also
	Project, Registration Database


S4G759. Readings on ProgID.

ProgID

The Programmatic Identifier (known as ProgID) is the string you pass to CreateObject() or NewObject() to create a COM object. This property tells you what string to use for a particular COM object you're creating. The property is read-only.

	Usage
	cMyServername = oServer.ProgID


When you create a COM server as part of a VFP Project, the server's name is created in the form "Project.ClassName," where Project is the name of your project and ClassName is the name of the visual class or program that defines the server class. To create an instance of your object, pass this name to CreateObject(). The ProgID displays the name to pass. 

You cannot change this name directly. You can change the first half by modifying the server name using the Project's ServerProject property, and the second half by renaming the class library or program that defines the object.

	Example
	* Ted's project for testing SYS(2335) was called SYS2335

* His test class was a program named Fred.PRG

? oProject.Servers[1].ProgID          && "SYS2335.fred"

oDear = CREATEOBJECT("SYS2335.Fred")  && creates the object


	See Also
	CreateObject(), NewObject(), Project, Server, ServerProject



